// File: contracts/rbn-staking/FeeCustody.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.11;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "../interfaces/IFeeDistributor.sol";
import "../interfaces/IChainlink.sol";
import "../interfaces/IWETH.sol";
import "../interfaces/IWSTETH.sol";
import "../interfaces/ICRV.sol";
import "../interfaces/ISwapRouter.sol";

import "@uniswap/v3-periphery/contracts/libraries/TransferHelper.sol";

/** @title FeeCustody
    @notice Custody Contract for Ribbon Vault Management / Performance Fees
 */

contract FeeCustody is Ownable {
  using SafeERC20 for IERC20;

  address public constant WETH = 0xC02aaA39b223FE8D0A0e5C4F27eAD9083C756Cc2;
  address public constant WSTETH = 0x7f39C581F595B53c5cb19bD0b3f8dA6c935E2Ca0;
  address public constant STETH = 0xae7ab96520DE3A18E5e111B5EaAb095312D7fE84;

  // WETH Distribution Token
  IERC20 public distributionToken = IERC20(WETH);
  // Protocol revenue recipient
  address public protocolRevenueRecipient;
  // Address of fee distributor contract for RBN lockers to claim
  IFeeDistributor public feeDistributor;

  // % allocation (0 - 100%) from protocol revenue to allocate to RBN lockers.
  // 2 decimals. ex: 10% = 1000
  uint256 public pctAllocationForRBNLockers;

  uint256 public constant TOTAL_PCT = 10000; // Equals 100%
  ISwapRouter public constant UNIV3_SWAP_ROUTER =
    ISwapRouter(0x68b3465833fb72A70ecDF485E0e4C7bD8665Fc45);
  ICRV public constant STETH_ETH_CRV_POOL =
    ICRV(0xDC24316b9AE028F1497c275EB9192a3Ea0f67022);

  // Intermediary path asset for univ3 swaps.
  // Empty if direct pool swap between asset and distribution asset
  mapping(address => bytes) public intermediaryPath;

  // Oracle between asset/usd pair for total
  // reward approximation across all assets earned
  mapping(address => address) public oracles;

  address[] public assets;

  // Keeper for weekly distributions
  address public keeper;

  // Events
  event NewAsset(address asset, bytes intermediaryPath);
  event RecoveredAsset(address asset);
  event NewFeeDistributor(address feeDistributor);
  event NewRBNLockerAllocation(uint256 pctAllocationForRBNLockers);
  event NewDistributionToken(address distributionToken);
  event NewProtocolRevenueRecipient(address protocolRevenueRecipient);
  event NewKeeper(address keeper);

  /**
   * @notice
   * Constructor
   * @param _pctAllocationForRBNLockers percent allocated for RBN lockers (100% = 10000)
   * @param _feeDistributor address of fee distributor where protocol revenue claimable
   * @param _protocolRevenueRecipient address of multisig
   * @param _admin admin
   * @param _keeper keeper
   */
  constructor(
    uint256 _pctAllocationForRBNLockers,
    address _feeDistributor,
    address _protocolRevenueRecipient,
    address _admin,
    address _keeper
  ) {
    require(_feeDistributor != address(0), "!_feeDistributor");
    require(
      _protocolRevenueRecipient != address(0),
      "!_protocolRevenueRecipient"
    );
    require(_admin != address(0), "!_admin");
    require(_keeper != address(0), "!_keeper");

    pctAllocationForRBNLockers = _pctAllocationForRBNLockers;
    feeDistributor = IFeeDistributor(_feeDistributor);
    protocolRevenueRecipient = _protocolRevenueRecipient;
    keeper = _keeper;
  }

  receive() external payable {}

  /**
   * @notice
   * Swaps RBN locker allocation of protocol revenu to distributionToken,
   * sends the rest to the multisig
   * @dev Can be called by keeper
   * @param _minAmountOut min amount out for every asset type swap.
   * will need to be in order of assets in assets[] array. should be
   * fine if we keep track.
   * @return toDistribute amount of distributionToken distributed to fee distributor
   */
  function distributeProtocolRevenue(uint256[] calldata _minAmountOut)
    external
    returns (uint256 toDistribute)
  {
    require(msg.sender == keeper, "!keeper");

    if (address(distributionToken) == WETH) {
      IWETH(address(distributionToken)).deposit{value: address(this).balance}();
    }

    uint256 len = assets.length;
    for (uint256 i; i < len; i++) {
      IERC20 asset = IERC20(assets[i]);
      uint256 assetBalance = asset.balanceOf(address(this));

      if (assetBalance == 0) {
        continue;
      }

      uint256 multiSigRevenue = (assetBalance *
        (TOTAL_PCT - pctAllocationForRBNLockers)) / TOTAL_PCT;

      // If we are holding the distributionToken itself,
      // do not swap
      if (address(asset) != address(distributionToken)) {
        // Calculate RBN allocation amount to swap for distributionToken
        uint256 amountIn = assetBalance - multiSigRevenue;
        _swap(address(asset), amountIn, _minAmountOut[i]);
      }

      // Transfer multisig allocation of protocol revenue to multisig
      asset.safeTransfer(protocolRevenueRecipient, multiSigRevenue);
    }

    toDistribute = distributionToken.balanceOf(address(this));
    distributionToken.safeApprove(address(feeDistributor), toDistribute);

    // Tranfer RBN locker allocation of protocol revenue to fee distributor
    feeDistributor.burn(address(distributionToken), toDistribute);
  }

  /**
   * @notice
   * Amount of _asset allocated to RBN lockers from current balance
   * @return amount allocated to RBN lockers
   */
  function claimableByRBNLockersOfAsset(address _asset)
    external
    view
    returns (uint256)
  {
    uint256 allocPCT = pctAllocationForRBNLockers;
    uint256 balance = _asset == address(0)
      ? address(this).balance
      : IERC20(_asset).balanceOf(address(this));
    return (balance * allocPCT) / TOTAL_PCT;
  }

  /**
   * @notice
   * Amount of _asset allocated to multisig from current balance
   * @return amount allocated to multisig
   */
  function claimableByProtocolOfAsset(address _asset)
    external
    view
    returns (uint256)
  {
    uint256 allocPCT = TOTAL_PCT - pctAllocationForRBNLockers;
    uint256 balance = _asset == address(0)
      ? address(this).balance
      : IERC20(_asset).balanceOf(address(this));
    return (balance * allocPCT) / TOTAL_PCT;
  }

  /**
   * @notice
   * Total allocated to RBN lockers across all assets balances
   * @return total allocated (in USD)
   */
  function totalClaimableByRBNLockersInUSD() external view returns (uint256) {
    uint256 allocPCT = pctAllocationForRBNLockers;
    return _getTotalAssetValue(allocPCT);
  }

  /**
   * @notice
   * Total allocated to multisig across all assets balances
   * @return total allocated (in USD)
   */
  function totalClaimableByProtocolInUSD() external view returns (uint256) {
    uint256 allocPCT = TOTAL_PCT - pctAllocationForRBNLockers;
    return _getTotalAssetValue(allocPCT);
  }

  /**
   * @notice
   * Total claimable across all asset balances based on allocation PCT
   * @param _allocPCT allocation percentage
   * @return claimable total claimable (in USD)
   */
  function _getTotalAssetValue(uint256 _allocPCT)
    internal
    view
    returns (uint256 claimable)
  {
    uint256 len = assets.length;
    for (uint256 i; i < len; i++) {
      IChainlink oracle = IChainlink(oracles[assets[i]]);

      ERC20 asset = ERC20(assets[i]);

      uint256 balance = asset.balanceOf(address(this)) *
        (10**(18 - asset.decimals()));

      if (assets[i] == WETH) {
        balance += address(this).balance;
      }

      // Approximate claimable by multiplying
      // current asset balance with current asset price in USD
      claimable +=
        (balance * uint256(oracle.latestAnswer()) * _allocPCT) /
        10**8 /
        TOTAL_PCT;
    }
  }

  /**
   * @notice
   * Swaps _amountIn of _asset into distributionToken
   * @param _asset asset to swap from
   * @param _amountIn amount to swap of asset
   * @param _minAmountOut min amount out for every asset type swap
   */
  function _swap(
    address _asset,
    uint256 _amountIn,
    uint256 _minAmountOut
  ) internal {
    if (_asset == WSTETH) {
      uint256 _stethAmountIn = IWSTETH(_asset).unwrap(_amountIn);
      TransferHelper.safeApprove(
        STETH,
        address(STETH_ETH_CRV_POOL),
        _stethAmountIn
      );

      STETH_ETH_CRV_POOL.exchange(1, 0, _stethAmountIn, _minAmountOut);

      IWETH(address(distributionToken)).deposit{value: address(this).balance}();
      return;
    }

    TransferHelper.safeApprove(_asset, address(UNIV3_SWAP_ROUTER), _amountIn);

    ISwapRouter.ExactInputParams memory params = ISwapRouter.ExactInputParams({
      path: intermediaryPath[_asset],
      recipient: address(this),
      amountIn: _amountIn,
      amountOutMinimum: _minAmountOut
    });

    // Executes the swap.
    UNIV3_SWAP_ROUTER.exactInput(params);
  }

  /**
     * @notice
     * add asset
     * @dev Can be called by admin
     * @param _asset new asset
     * @param _oracle ASSET/USD ORACLE.
     * @param _intermediaryPath path for univ3 swap.
     * @param _poolFees fees for asset / distributionToken.

     * If intermediary path then pool fee between both pairs
     * (ex: AAVE / ETH , ETH / USDC)
     * NOTE: if intermediaryPath empty then single hop swap
     * NOTE: MUST BE ASSET / USD ORACLE
     * NOTE: 3000 = 0.3% fee for pool fees
     */
  function setAsset(
    address _asset,
    address _oracle,
    address[] calldata _intermediaryPath,
    uint24[] calldata _poolFees
  ) external onlyOwner {
    require(_asset != address(0), "!_asset");
    uint256 _pathLen = _intermediaryPath.length;
    uint256 _swapFeeLen = _poolFees.length;

    // We must be setting new valid oracle, or want to keep as is if one exists
    require(IChainlink(_oracle).decimals() == 8, "!ASSET/USD");
    require(_pathLen < 2, "invalid intermediary path");
    require(_swapFeeLen == _pathLen + 1, "invalid pool fees array length");

    // If not set asset
    if (oracles[_asset] == address(0)) {
      assets.push(_asset);
    }

    // Set oracle for asset
    oracles[_asset] = _oracle;

    // Multiple pool swaps are encoded through bytes called a `path`.
    // A path is a sequence of token addresses and poolFees that define the pools used in the swaps.
    // The format for pool encoding is (tokenIn, fee, tokenOut/tokenIn, fee, tokenOut)
    // where tokenIn/tokenOut parameter is the shared token across the pools.
    if (_pathLen > 0) {
      intermediaryPath[_asset] = abi.encodePacked(
        _asset,
        _poolFees[0],
        _intermediaryPath[0],
        _poolFees[1],
        address(distributionToken)
      );
    } else {
      intermediaryPath[_asset] = abi.encodePacked(
        _asset,
        _poolFees[0],
        address(distributionToken)
      );
    }

    emit NewAsset(_asset, intermediaryPath[_asset]);
  }

  /**
   * @notice
   * recover all assets
   * @dev Can be called by admin
   */
  function recoverAllAssets() external onlyOwner {
    // For all added assets, send to protocol revenue recipient
    uint256 len = assets.length;
    for (uint256 i = 0; i < len; i++) {
      _recoverAsset(assets[i]);
    }
  }

  /**
   * @notice
   * recover specific asset
   * @dev Can be called by admin
   * @param _asset asset to recover
   */
  function recoverAsset(address _asset) external onlyOwner {
    require(_asset != address(0), "!asset");
    _recoverAsset(_asset);
  }

  /**
   * @notice
   * recovers asset logic
   * @param _asset asset to recover
   */
  function _recoverAsset(address _asset) internal {
    IERC20 asset = IERC20(_asset);
    uint256 bal = asset.balanceOf(address(this));
    if (bal > 0) {
      asset.safeTransfer(protocolRevenueRecipient, bal);
      emit RecoveredAsset(_asset);
    }
  }

  /**
   * @notice
   * set fee distributor
   * @dev Can be called by admin
   * @param _feeDistributor new fee distributor
   */
  function setFeeDistributor(address _feeDistributor) external onlyOwner {
    require(_feeDistributor != address(0), "!_feeDistributor");
    feeDistributor = IFeeDistributor(_feeDistributor);
    emit NewFeeDistributor(_feeDistributor);
  }

  /**
   * @notice
   * set rbn locker allocation pct
   * @dev Can be called by admin
   * @param _pctAllocationForRBNLockers new allocation for rbn lockers
   */
  function setRBNLockerAllocPCT(uint256 _pctAllocationForRBNLockers)
    external
    onlyOwner
  {
    require(
      _pctAllocationForRBNLockers <= TOTAL_PCT,
      "!_pctAllocationForRBNLockers"
    );
    pctAllocationForRBNLockers = _pctAllocationForRBNLockers;
    emit NewRBNLockerAllocation(_pctAllocationForRBNLockers);
  }

  /**
   * @notice
   * set new distribution asset
   * @dev Can be called by admin
   * @param _distributionToken new distribution token
   */
  function setDistributionToken(address _distributionToken) external onlyOwner {
    require(_distributionToken != address(0), "!_distributionToken");
    distributionToken = IERC20(_distributionToken);
    emit NewDistributionToken(_distributionToken);
  }

  /**
   * @notice
   * set protocol revenue recipient
   * @dev Can be called by admin
   * @param _protocolRevenueRecipient new protocol revenue recipient
   */
  function setProtocolRevenueRecipient(address _protocolRevenueRecipient)
    external
    onlyOwner
  {
    require(
      _protocolRevenueRecipient != address(0),
      "!_protocolRevenueRecipient"
    );
    protocolRevenueRecipient = _protocolRevenueRecipient;
    emit NewProtocolRevenueRecipient(_protocolRevenueRecipient);
  }

  /**
   * @notice
   * set keeper for weekly distributions
   * @dev Can be called by admin
   * @param _keeper new keeper
   */
  function setKeeper(address _keeper) external onlyOwner {
    require(_keeper != address(0), "!_keeper");
    keeper = _keeper;
    emit NewKeeper(_keeper);
  }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `recipient` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * Requirements:
     *
     * - `sender` and `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     * - the caller must have allowance for ``sender``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) public virtual override returns (bool) {
        _transfer(sender, recipient, amount);

        uint256 currentAllowance = _allowances[sender][_msgSender()];
        require(currentAllowance >= amount, "ERC20: transfer amount exceeds allowance");
        unchecked {
            _approve(sender, _msgSender(), currentAllowance - amount);
        }

        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender] + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        uint256 currentAllowance = _allowances[_msgSender()][spender];
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(_msgSender(), spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `sender` to `recipient`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `sender` cannot be the zero address.
     * - `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     */
    function _transfer(
        address sender,
        address recipient,
        uint256 amount
    ) internal virtual {
        require(sender != address(0), "ERC20: transfer from the zero address");
        require(recipient != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(sender, recipient, amount);

        uint256 senderBalance = _balances[sender];
        require(senderBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[sender] = senderBalance - amount;
        }
        _balances[recipient] += amount;

        emit Transfer(sender, recipient, amount);

        _afterTokenTransfer(sender, recipient, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: contracts/interfaces/IFeeDistributor.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.11;

interface IFeeDistributor{
    function burn(address coin, uint256 amount) external returns (bool);
}

// File: contracts/interfaces/IChainlink.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity >=0.7.2;
pragma experimental ABIEncoderV2;

interface IChainlink {
  function decimals() external view returns (uint8);

  function latestAnswer() external view returns (int256);
}

// File: contracts/interfaces/IWETH.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface IWETH {
    function deposit() external payable;

    function balanceOf(address account) external view returns (uint256);
}

// File: contracts/interfaces/IWSTETH.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface IWSTETH {
  function unwrap(uint256 _amount) external returns (uint256);
}

// File: contracts/interfaces/ICRV.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface ICRV {
  // https://github.com/curvefi/curve-contract/blob/
  // b0bbf77f8f93c9c5f4e415bce9cd71f0cdee960e/contracts/pools/steth/StableSwapSTETH.vy#L431
  function exchange(
    int128 _indexIn,
    int128 _indexOut,
    uint256 _amountIn,
    uint256 _minAmountOut
  ) external payable returns (uint256);
}

// File: contracts/interfaces/ISwapRouter.sol



// SPDX-License-Identifier: GPL-2.0-or-later
pragma solidity ^0.8.0;

/// @title Router token swapping functionality
/// @notice Functions for swapping tokens via Uniswap V3
interface ISwapRouter {
  struct ExactInputParams {
    bytes path;
    address recipient;
    uint256 amountIn;
    uint256 amountOutMinimum;
  }

  /// @notice Swaps `amountIn` of one token for as much as possible of another along the specified path
  /// @param params The parameters necessary for the multi-hop swap, encoded as `ExactInputParams` in calldata
  /// @return amountOut The amount of the received token
  function exactInput(ExactInputParams calldata params) external payable returns (uint256 amountOut);
}

// File: @uniswap/v3-periphery/contracts/libraries/TransferHelper.sol



// SPDX-License-Identifier: GPL-2.0-or-later
pragma solidity >=0.6.0;

import '@openzeppelin/contracts/token/ERC20/IERC20.sol';

library TransferHelper {
    /// @notice Transfers tokens from the targeted address to the given destination
    /// @notice Errors with 'STF' if transfer fails
    /// @param token The contract address of the token to be transferred
    /// @param from The originating address from which the tokens will be transferred
    /// @param to The destination address of the transfer
    /// @param value The amount to be transferred
    function safeTransferFrom(
        address token,
        address from,
        address to,
        uint256 value
    ) internal {
        (bool success, bytes memory data) =
            token.call(abi.encodeWithSelector(IERC20.transferFrom.selector, from, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'STF');
    }

    /// @notice Transfers tokens from msg.sender to a recipient
    /// @dev Errors with ST if transfer fails
    /// @param token The contract address of the token which will be transferred
    /// @param to The recipient of the transfer
    /// @param value The value of the transfer
    function safeTransfer(
        address token,
        address to,
        uint256 value
    ) internal {
        (bool success, bytes memory data) = token.call(abi.encodeWithSelector(IERC20.transfer.selector, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'ST');
    }

    /// @notice Approves the stipulated contract to spend the given allowance in the given token
    /// @dev Errors with 'SA' if transfer fails
    /// @param token The contract address of the token to be approved
    /// @param to The target of the approval
    /// @param value The amount of the given token the target will be allowed to spend
    function safeApprove(
        address token,
        address to,
        uint256 value
    ) internal {
        (bool success, bytes memory data) = token.call(abi.encodeWithSelector(IERC20.approve.selector, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'SA');
    }

    /// @notice Transfers ETH to the recipient address
    /// @dev Fails with `STE`
    /// @param to The destination of the transfer
    /// @param value The value to be transferred
    function safeTransferETH(address to, uint256 value) internal {
        (bool success, ) = to.call{value: value}(new bytes(0));
        require(success, 'STE');
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}